

Weedkiller



With a different gameplay game 51 would become a new game.
This game you kill the growing weed but once grown becomes impenetrable.
With the base of game 51 game 52 is coded in 1K.

```
; Game 52 in 1K hires : weedkiller
```

```
? * TORNADO *
```

```
dataline EQU #4000
```

```
ORG #4009 ;#4009  
DUMP 49161
```

```
fl EQU 6
```

```
basic LD D,#C0 ; preset for 48K bug  
JR init0 ; this game has no 48K bug
```

```
DEFB 236,212,28 ; The BASIC  
DEFB 126 ; fully placed over sysvar  
DEFB 143,0,18 ; start to BASIC=#4009
```

```
eline DEFW last ; needed to load
```

```
chadd DEFW last-1
```

```
xptr DEFW 0
```

```
stkbot DEFW last ; needed to load
```

```
stkend DEFW last ; needed to load
```

```
berg DEFB 0
```

```
mem DEFW 0
```

```

        DEFB 0                      ; 128

initl1   JP    init                ; init can be anywhere

; all above reusable AFTER loading

lastk    DEFB 255,255,255          ; used by ZX81
margin   DEFB 55                  ; used by ZX81
nxtlin    DEFW basic              ; reusable after load

init0     LD    E,L                ; delay intrupts by
          DEFB #26                ; LD H,64
flagx     DEFB 64                 ; clever setting of flags

          XOR    A                ; intruptcounter reset
          EX     AF,AF'

taddr     DEFW 0                  ; used by ZX81,no hurting code
dead      LD    B,4               ; frames is set ok

frames    DEFW #DD*256+1          ; used by ZX81, clever IX set
coprcc    LD    HL,hr             ; set IX
sposn     JR    initl
cdflag    DEFB 64                ; used by zx81

udg4      EQU    bush*256/256
udg2      EQU    weed*256/256
udg3      EQU    kill*256/256
udg1      EQU    spc*256/256

bush      DEFB 0,96,0,3,152,0,12,68,0
          DEFB 56,34,0,196,135,0,129,49,128
          DEFB 98,72,128
          DEFB 16,68,240,16,0,136,98,128,144
          DEFB 133,65,144,133,65,144,136,34,36
          DEFB 129,8,42,66,148,50,64,0,4
          DEFB 127,255,248

kill       DEFB 0,0,160,0,0,160,0,0,64
          DEFB 0,0,64,0,0,64,0,0,64
          DEFB 0,0,160,0,0,160,0,0,160
          DEFB 0,0,64,0,0,64,0,0,64
          DEFB 0,0,160,0,0,160

spc        DEFW 0,0,0,0,0,0,0,0
          DEFW 0,0,0,0,0,0,0,0

          DEFB 128,0,1
weed        DEFB 0
          DEFW 0,0,0,0,0,0,0,0,0,0
          DEFB 0,8,0,0,148,0,1,72,64
          DEFB 16,148,164,40,120,74,48,192,68
          DEFB 20,64,44,8,64,44
          DEFB 8,64,68,127,255,254

lbuf2      LD    R,A
          DEFW 0,0,0,0,0,0,0,0,0,0,0
          JP     (HL)

hr         LD    HL,lowres+#8000   ; the lowres display
          LD    BC,#231           ; minimum needed
          LD    A,#1E

```

```

LD    I,A
LD    A,#FB
CALL  #2B5

hr00   LD    B,4                ; sync hires display
      DJNZ  hr00
      NOP

      LD    HL,low              ; return from display
      EXX

      LD    A,#40
      LD    I,A                ; set high byte display

      LD    B,fl+1              ; the number of floors +1
      LD    (savesp+1),SP       ; save SP
      LD    SP,ixpoint          ; get floors from stack
      JR    cloop

loop1   POP   IX                ; get next floor
      LD    HL,udg1*256+udg2    ; window and fire
      LD    DE,udg3*256+udg4    ; water and cracked window
      LD    C,17                ; 16 lines per floor
      JR    nline

bloop   INC   DE                ; next line cracked window
      INC   E
      INC   E

      INC   D                  ; next line water
      INC   D
      INC   D

      INC   HL                ; next line fire
      INC   HL
      INC   HL

      INC   H                  ; next line window
      INC   H
      INC   H

nline   DEC   C
      LD    A,(HL)              ; filler
      JP    (IX)                ; show floor

low      EXX
      JP    Z,cloop             ; show next floor
      JP    bloop              ; show rest current floor

cloop   DJNZ  loop1             ; extra line

savesp   LD    SP,0             ; retrieve stack

hr01     LD    B,6
      LD    DE,poisonudg
      RET   NZ
      DJNZ  hr01

      LD    A,(HL)

      LD    BC,#11FF

poisonpos LD    A,dataline*256/256

```

```

poisonx      EX    DE,HL
              LD    DE,dataline
              LDI
              LDI
              LDI
              EX    DE,HL
              LD    HL,low3
              JP    lbuf2+#8000

low3          DJNZ  poisonpos

              CALL  #292                ; back from intrupt
              CALL  #220
              LD    IX,hr
              JP    #2A4

poisonudg    DEFB  63,248,64,64,4,64,143,226,64
              DEFB  143,226,64,144,18,64,148,82,64
              DEFB  144,18,64,139,162,64,171,170,64
              DEFB  152,50,224,167,202,160,128,2,160
              DEFB  179,58,160
              DEFB  170,147,32,179,16,64,255,255,128
              DEFB  0,0,0

ixpoint      DEFW  lbuf+32768            ; each floor
              DEFW  lbuf+32794            ; has own display
              DEFW  lbuf+32820            ; 6 floor building
              DEFW  lbuf+32846
              DEFW  lbuf+32872
              DEFW  lbuf+32898

flsize       EQU   26*fl-26

eog          LD    HL,score-1
              LD    DE,hiscore-1
              LD    C,7                  ; b=0 here

fihi         INC    HL
              INC    DE
              DEC    C
              JR     Z,start
              LD     A,(DE)
              CP     (HL)
              JR     Z,fihi
              CALL   C,#19F9              ; set new high

start        LD     A,(lastk)             ; game over, wait for
              SUB    %10111111           ; newline
              JR     NZ,start
              LD     (dead),A

              LD     L,score*256/256

ressc        LD     (HL),28              ; reset score
              INC    HL
              CP     (HL)
              JR     NZ,ressc

              LD     A,14
              LD     (taddr),A

```

```

LD    L,lbuff*256/256+24+3
LD    DE,65535-5          ; -6
LD    BC,1024+flsize
mkcopy ADD HL,DE
LD    (HL),#3C            ; repair all rooms
DJNZ  mkcopy

LD    L,lbuff*256/256     ; all floors the same
LD    DE,init
LDIR

CALL  showall             ; show floors

nextroom LD    B,fl
CALL  rnd
LD    C,A                 ; set floor

CALL  rnd
AND   3
INC   A
LD    B,A                 ; set app on floor
LD    (fircol+1),A        ; save firecolumn

CALL  field
CP    #7B                 ; bush test
JR    Z,nextroom

LD    (HL),#7D            ; let weed grow
LD    (fireapp+1),HL      ; save app location

LD    B,20
CALL  rnd
ADD   A,B
ADD   A,B
LD    (dead+1),A          ; set timer

playloop JR    Z,eog

LD    A,%11011111        ; Y-P
IN    A,(254)
LD    HL,poisonx+1
LD    C,(HL)
RRA                     ; P
JR    C,t2
INC   (HL)
t2     RRA                 ; 0
JR    C,t3
DEC   (HL)
t3     LD    A,(HL)
CP    19
JR    C,colcheck         ; not out of line
LD    (HL),C             ; undo illegal move

colcheck LD    A,(HL)      ; get xpos
fircol   LD    B,0
LD    C,B
ADD   A,6
colnr    SUB   6
DJNZ  colnr
JR    NZ,delaylp         ; not in column

LD    B,C
LD    A,(lastk)
LD    HL,power

```

```

SUB %11111011          ; Q-T
JR  Z,release

CP  %11111101-%11111011 ; A-G
JR  NZ,drive

old    CP  0              ; keyrepeat, is no pumping
LD  B,A
JR  Z,setold

LD  A,(HL)
SUB  28+fl
JR  Z,setold
INC  (HL)                ; increase waterpressure

setold LD  A,B
drive  LD  (old+1),A      ; signal other key pressed

delaylp CALL #4016       ; wait some time

LD  HL,dead+1           ; decrease timer
DEC  (HL)
JR  NZ,pl
DEC  HL
INC  (HL)                ; weed grows into bush
LD  A,(HL)
SUB  3
LD  HL,(fireapp+1)
INC  HL
INC  HL
INC  HL
LD  (HL),#3B             ; set signal bush
PUSH AF
CALL showall            ; show fields
LD  HL,taddr
DEC  (HL)
JR  NZ,cont
POP  DE
DEFB 62
cont   POP  AF
JR  NZ,nr                ; not eog, get next weed

pl     JP  playloop      ; continue test eog

release LD  C,A           ; C now 0
LD  A,(HL)
SUB  28
JR  Z,delaylp

showwater CALL field      ; get lowest field
CP  #7B
LD  D,(HL)              ; read what is it
JR  Z,stopper           ; bush stops spray
LD  (HL),#7A            ; show DDT-spray
LD  A,(power)
SUB  C
INC  C
CP  28
JR  NZ,showwater        ; do full pressure

stopper PUSH DE           ; save latest appartement
LD  A,256-6

```

```

CALL #4018

CALL showall

fireapp    LD    HL,0
           LD    (HL),#7D          ; set weed back on
           POP   AF                ; test weed on field
           CP    #7D              ; weed killed      ?
           JR    NZ,drive

           LD    (HL),#7C          ; kill weed

; score = floor * remaining time
addc        LD    A,(dead+1)      ; get timer
           LD    B,A
addb        LD    HL,score+6
           DEFB  17
ten         LD    (HL),28
           DEC   HL
           INC   (HL)
           LD    A,(HL)
           CP    38
           JR    Z,ten
           DJNZ  addb
           DEC   C
           JR    NZ,addc
           XOR   A
           LD    (dead),A

nr          JP    nextroom        ; set next weed

showall     LD    B,24            ; show all original fields
           LD    HL,lbuf
allfield    LD    D,H            ; destination
           LD    E,L
           INC   HL
           INC   HL
           INC   HL
           LD    A,(HL)          ; get save location
           SET   6,A             ; make it LD A,r
           LD    (DE),A
aok         INC   HL
hl2         INC   HL
           INC   HL
           LD    A,(HL)          ; test end of line
           CP    #D9
           JR    Z,hl2
           DJNZ  allfield
           LD    L,power*256/256  ; reset pressure
           LD    (HL),28
           RET

field       LD    HL,lbuf-6-26
           LD    A,fl+1
           SUB   C
           LD    DE,26
flbuf       ADD   HL,DE
           DEC   A
           JR    NZ,flbuf
           LD    E,6
           LD    A,B
ffield     ADD   HL,DE
           DEC   A

```

```

        JR    NZ,ffield
        LD    A,(HL)
        RET

rnd      LD    HL,(frames)          ; read RND from ROM
rseed    LD    DE,0
        ADD   HL,DE
        DEC   HL
        LD    A,H
        AND   #1F
        LD    H,A
        LD    (rseed+1),HL
        LD    A,(HL)
frnd     SUB   B
        JR    NC,frnd
        ADC   A,B
        RET

x        EQU   101

lowres   DEFB  118
score    DEFB  28,28,28,28,28,28,0

        DEFB  "W"+x,"E"+x,"E"+x,"D"+x,"K"+x,"I"+x
        DEFB  "L"+x,"L"+x,"E"+x,"R"+x,0
power    DEFB  28,0
hiscore  DEFB  28,28,28,28,33,29
        DEFB  118

scsize   EQU   fl*26

space    EQU   #4400-26-scsize-$

        DEFS  space                ; no space left

lbuf     LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0
        EXX
        JP    (HL)

init     LD    SP,#4400
        LDIR                    ; repair 48K bug, DEFAULT

        LD    HL,delay
        LD    DE,#4016
        LD    C,10
        LDIR

mkline   LD    HL,#4016          ; built floor line
        DEC   L

```



```

LD      (HL),0
JR      NZ,mkline      ; built horizontal line

LD      HL,eog          ; start on end of game
PUSH    HL
LD      HL,lbuf
LD      DE,init
LD      BC,26*fl-26     ; make buildingfloors
JP      #19F9

delay   LD      A,256-3
        LD      HL,frames
        ADD     A,(HL)
wfr     CP      (HL)
        JR      NZ,wfr
        RET

vars    DEFB 128
?
last    EQU     $

```